# A larger 2000x1000 general LP (lp2000)

**Problem**: This problem is the general LP (linear program) with 2000 variables and 1000 linear constraints. The general formulation in vector form is as follows.

$$\begin{array}{ll} \max & \mathbf{c}' \cdot \mathbf{x} \\ \text{subject to} & \mathbf{A} \cdot \mathbf{x} \leq \mathbf{b} \\ & \mathbf{x} \geq 0 \end{array}$$

An explicit formulation can be given as follows:

$$\begin{array}{llll} \max & \sum_{j \in J} c_j x_j & & \\ \text{subject to} & \sum_{j \in J} a_{i,j} x_j \leq b_i & & \text{forall } i \in I \\ & x_j \geq 0 & & \text{forall } j \in J \\ & J = \{1, \ldots, m\}, & I = \{1, \ldots, n\} & m, n \geq 0 \end{array}$$

Model this problem as an LPL model. The data for the model are generated randomly and the fill-factor of the matrix **A** is about 2%.

**Modeling Steps**:

1. The first step is to add two sets $i \in I = \{\{1, \ldots, 1000\}\}$ and $j \in J = \{\{1, \ldots, 2000\}\}$. In LPL this is

```
set  i :=  1..1000;    j :=  1..2000;
```

2. Next the data tables **A**, **b**, and **c** are defined and assigned. `Rnd(0,1)` is a function that returns a uniform distributed number between in the interval [1, 1[. The function `if(C,E)` returns `E` if `C` is true otherwise it returns zero. Hence, the table **A** is a sparse table that contains only 2% of the entries (because zero entries are not stored in this case).

3. Next the variable vector is declared (declaring 2000 variables):

```
  variable  x{j};
```

4. Next the constraints are defined "`R{i}:`" indicates that 1000 constraints are declared. The expression "`sum{j} A*x <= b;`" is a shortcut for :

```
   sum{j} A[i,j]*x[j] <= b[i];
```

   (Indexes can be leaved out if the context allows it.)

5. Finally, the objective function value is written and a list of all variables with its value is wriiten each on a newline if the value is different from 0 in a formatted way.

```
Write('Objective␣Value␣=␣%7.2f␣\n', Obj);
Write{j|x}('␣␣␣x%-4s␣=␣%6.2f\n' , j,x);
```

Listing 1: The Complete Model implemented in LPL [2]

```
model Lp2000 "A larger 2000x1000 general LP (math)";
  set  i :=  1..1000;    j :=  1..2000;
  parameter
    A{i,j} := if(Rnd(0,1)<0.02 , Rnd(0,60));
    c{j}   := if(Rnd(0,1)<0.87 , Rnd(0,9));
    b{i}   := if(Rnd(0,1)<0.87 , Rnd(10,70000));
  variable  x{j};
  constraint R{i}:  sum{j} A*x <= b;
  maximize Obj:  sum{j}  c*x;
  Write('Objective␣Value␣=␣%7.2f␣\n', Obj);
  Write{j|x}('␣␣␣x%-4s␣=␣%6.2f\n' , j,x);
end
```

**Solution**: The solution output is as follows:

```
Objective Value = 77599.07
   x22    =  34.64
   x31    =  14.46
   ... 85 more values ...
   x1946 = 340.90
   x1986 =  77.13
```

**Further Comments**: Click the link `lp2000` at the top of this document to run the model in the Internet.

# References


[1] MatMod. Homepage for Learning Mathematical Modeling : https://matmod.ch.

[2] Hürlimann T. Reference Manual for the LPL Modeling Language, most recent version. https://matmod.ch/lpl/doc/manual.pdf.